
1. Overview

This document lists a TTCN change needed to correct issues in the ATS ‘iwd-EUTRA-B2012-03_D13wk08’.

3. Corrections

Change 1 – Renaming Condition Types in function ‘f_SelectionExpr ()’
	Function name
	f_ SelectionExpr()

	Reason for change
	Duplication of Condition Type names “Cxx” used in different sub-ATS within the “iwd-EUTRA-B2012-03_D13wk08” delivery caused misunderstandings and potential mistakes in overall Test Campaign Management.

	Summary of change
	Added an ATS-prefix to the Condition Type names in Selection Expressions modules of sub-ATS “IMS_IRAT” and “IMS_CC” (… and maybe also “LTE_EPS”) according to the naming convention introduced with “POS” and “UTRAN34”.

	TTCN module
	IMS_IRAT_TS_SelectionExpressions.ttcn

Similar renaming of Condition Types shall be done in TTCN module 
IMS_CC_TS_SelectionExpressions.ttcn

… and may be done in TTCN module 
LTE_EPS_TS_SelectionExpressions.ttcn 

	MCC160 Comment
	Current TTCN implementation is correct. Nevertheless as improvement, the conditions can be renamed to have a prefix per specification and not per test suite; a prefix per test suite would imply duplications and additional maintenance. 
Further optimisation can be done in the future depending on test suite split and the result of the discussion for harmonization of PICS at RAN5.

As a first step the following improvements are proposed:
· define only one module for the 34.229-2 selection expression  IMS_34229_SelectionExpressions. This module is used in IMS_IRAT and IMS_CC test suites.  
· rename the selection expression with the prefix IMS34


Before change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2013, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TTA, TTC).
//   All rights reserved.
// @version: IMS_36523_IRAT_IWD_13wk08
// $Date: 2013-02-14 19:25:52 +0100 (Thu, 14 Feb 2013) $
// $Rev: 8088 $
/******************************************************************************/
module IMS_IRAT_TS_SelectionExpressions {
  import from IMS_CC_Parameters all;
  import from UTRAN_Parameters all;
  import from Parameters all;
  type enumerated Condition_Type {
    R,
    C59,
    C62,
    C64,
    C66
  };
  function f_SelectionExpr ( Condition_Type p_Condition) return boolean
  {
    /* Set the applicability conditions according to 34.229-2 clause 4 table "Conditions/Options" */
    var boolean v_ApplCond;
    select (p_Condition) {
      case (R) {
        v_ApplCond := true;
      }
    case (C59) {
        // IMS emergency services
        v_ApplCond := pc_IMS_EmergencyCall;
      }
    case (C62) {
        //IMS emergency services and emergency speech call
        v_ApplCond := pc_IMS_EmergencyCall and pc_EmergSpeech ;
      }
    case (C64) {
        //IMS emergency services and EUTRA FDD or EUTRA TDD
        v_ApplCond := pc_IMS_EmergencyCall and (pc_eFDD or pc_eTDD);
      }
    case (C66) {
        //Speech and IMS emergency call and emergency call over GERAN or UTRAN or 1XRTT
        v_ApplCond := pc_BidirecVoiceOverIMS and pc_MTSI_Speech and (pc_EmergSpeech or pc_1xRTT) ;
    }
    case else {
        v_ApplCond := false;
      }
    }
    return v_ApplCond;
  }
}
…
After change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2013, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TTA, TTC).
//   All rights reserved.
// @version: IMS_36523_IRAT_IWD_13wk08
// $Date: 2013-02-14 19:25:52 +0100 (Thu, 14 Feb 2013) $
// $Rev: 8088 $
/******************************************************************************/
module IMS_IRAT_TS_SelectionExpressions {
  import from IMS_CC_Parameters all;
  import from UTRAN_Parameters all;
  import from Parameters all;
  type enumerated Condition_Type {
    R,
    IMS_IRAT_C59,
    IMS_IRAT_C62,
    IMS_IRAT_C64,
    IMS_IRAT_C66
  };
  function f_SelectionExpr ( Condition_Type p_Condition) return boolean
  {
    /* Set the applicability conditions according to 34.229-2 clause 4 table "Conditions/Options" */
    var boolean v_ApplCond;
    select (p_Condition) {
      case (R) {
        v_ApplCond := true;
      }
    case (IMS_IRAT_C59) {
        // IMS emergency services
        v_ApplCond := pc_IMS_EmergencyCall;
      }
    case (IMS_IRAT_C62) {
        //IMS emergency services and emergency speech call
        v_ApplCond := pc_IMS_EmergencyCall and pc_EmergSpeech ;
      }
    case (IMS_IRAT_C64) {
        //IMS emergency services and EUTRA FDD or EUTRA TDD
        v_ApplCond := pc_IMS_EmergencyCall and (pc_eFDD or pc_eTDD);
      }
    case (IMS_IRAT_C66) {
        //Speech and IMS emergency call and emergency call over GERAN or UTRAN or 1XRTT
        v_ApplCond := pc_BidirecVoiceOverIMS and pc_MTSI_Speech and (pc_EmergSpeech or pc_1xRTT) ;
    }
    case else {
        v_ApplCond := false;
      }
    }
    return v_ApplCond;
  }
}
…  

MCC160 proposed implementation
module IMS_34229_SelectionExpressions {

  import from IMS_CommonParameters all;

  import from IMS_CC_Parameters all;

  import from UTRAN_Parameters all;

  import from Parameters all;

  type enumerated Condition_Type {

    IMS34_R,

    IMS34_C04,

    IMS34_C05,

    IMS34_C06,

    IMS34_C07,

    IMS34_C08,

    IMS34_C09,

    IMS34_C10,

    IMS34_C17,

    IMS34_C18,

    IMS34_C19,

    IMS34_C22,

    IMS34_C23,

    IMS34_C26,

    IMS34_C27,

    IMS34_C28,

    IMS34_C31,

    IMS34_C32,

    IMS34_C37,

    IMS34_C43,

    IMS34_C44,

    IMS34_C48,

    IMS34_C49,

    IMS34_C50,

    IMS34_C55,

    IMS34_C56,

    IMS34_C57,

    IMS34_C58,

    IMS34_C59,

    IMS34_C62,

    IMS34_C63,

    IMS34_C64,

    IMS34_C65,

    IMS34_C66,

    IMS34_C68,

    IMS34_C69,

    IMS34_C70,

    IMS34_C71,

    IMS34_C72,

    IMS34_C73

  };

  function f_SelectionExpr ( Condition_Type p_Condition) return boolean

  {

    /* Set the applicability conditions according to 34.229-2 clause 4 table "Conditions/Options" */

    var boolean v_ApplCond;

    select (p_Condition) {

      case (IMS34_R) {

        v_ApplCond := true;

      }

      case (IMS34_C04) {

        // Dedicated PDP Context

        v_ApplCond := pc_InitiatesDedicatedPDPContext;

      }

      case (IMS34_C05) {

        // P-CSCF Discovery via PCO

        v_ApplCond := pc_P_CSCFDiscovery_viaPCO;

      }

      case (IMS34_C06) {

        // IPv4 AND configured to initiate P-CSCF discovery via DHCPv4

        v_ApplCond :=  pc_IPv4 and pc_InitiateP_CSCFDiscovery_viaDHCPv4;

      }

      case (IMS34_C07) {

        // IPv4 AND P-CSCF discovery via PCO AND P-CSCF discovery via DHCPv4 AND configured to initiate P-CSCF discovery via PCO

        v_ApplCond := pc_IPv4 and pc_P_CSCFDiscovery_viaPCO and pc_P_CSCFDiscovery_viaDHCPv4 and pc_InitiateP_CSCFDiscovery_viaPCO;

      }

      case (IMS34_C08) {

        // Configured to initiate P-CSCF discovery via DHCPv6

        v_ApplCond := pc_InitiateP_CSCFDiscovery_viaDHCPv6;

      }

      case (IMS34_C09) {

        // P-CSCF Discovery via PCO AND P-CSCF discovery via DHCPv6 AND configured to initiate P-CSCF discovery via PCO

        v_ApplCond := pc_P_CSCFDiscovery_viaPCO and pc_P_CSCFDiscovery_viaDHCPv6 and pc_InitiateP_CSCFDiscovery_viaPCO;

      }

      case (IMS34_C10) {

        //P-CSCF Discovery via PCO AND P-CSCF discovery via DHCPv6 AND configured to initiate P-CSCF discovery via DHCPv6

        v_ApplCond := pc_P_CSCFDiscovery_viaPCO and pc_P_CSCFDiscovery_viaDHCPv6 and pc_InitiateP_CSCFDiscovery_viaDHCPv6;

      }

      case (IMS34_C17) {

        // IMS security

        v_ApplCond := pc_IMS_Sec;

      }

      case (IMS34_C18) {

        // GIBA AND NOT IMS security

        v_ApplCond := pc_IMS_GIBA_Sec and not(pc_IMS_Sec);

      }

      case (IMS34_C19) {

        // IMS security AND GIBA

        v_ApplCond := pc_IMS_Sec and pc_IMS_GIBA_Sec;

      }

      case (IMS34_C22) {

        // MTSI and speech

        v_ApplCond := pc_MultimediaTelephonyService and  pc_MTSI_Speech;

      }

      case (IMS34_C23) {

        // MTSI and speech and Communication Hold

        v_ApplCond := pc_MultimediaTelephonyService and  pc_MTSI_Speech and pc_MTSI_CommHold ;

      }

      case (IMS34_C26) {

        // MTSI AND Initiating session AND preconditions AND MTSI text, RTP

        v_ApplCond := pc_MultimediaTelephonyService and pc_InitiateSession and pc_Preconditions and pc_MTSI_Text_RTP;

      }

      case (IMS34_C27) {

        // MTSI AND Initiating session AND MTSI speech

        v_ApplCond := pc_MultimediaTelephonyService and pc_InitiateSession and pc_MTSI_Speech;

      }

      case (IMS34_C28) {

        // MTSI AND Initiating session AND MTSI speech AND MTSI speech, AMR wideband

        v_ApplCond := pc_MultimediaTelephonyService and pc_InitiateSession and  pc_MTSI_Speech and pc_MTSI_Speech_AMRWB;

      }

      case (IMS34_C31) {

        // MTSI and speech and Communication Diversion

        v_ApplCond := pc_MultimediaTelephonyService and pc_MTSI_CommDivert and  pc_MTSI_Speech;

      }

      case (IMS34_C32) {

        // MTSI and Conference

        v_ApplCond := pc_MultimediaTelephonyService and pc_MTSI_Conference;

      }

      case (IMS34_C37) {

        //MTSI AND Initiating session AND MTSI text, RTP

        v_ApplCond := pc_MultimediaTelephonyService and pc_InitiateSession and pc_MTSI_Text_RTP;

      }

      case (IMS34_C43) {

        //UE supporting MTSI and Originating Identification Presentation

        v_ApplCond := pc_MultimediaTelephonyService and pc_MTSI_OIP;

      }

      case (IMS34_C44) {

        // UE supporting MTSI and Originating Identification Restriction

        v_ApplCond := pc_MultimediaTelephonyService and pc_MTSI_OIR;

      }

      case (IMS34_C48) {

        // MTSI and Terminating Identification Presentation

        v_ApplCond := pc_MultimediaTelephonyService and pc_MTSI_TIP;

      }

      case (IMS34_C49) {

        //MTSI and Terminating Identification Restriction

        v_ApplCond := pc_MultimediaTelephonyService and pc_MTSI_TIR;

      }

      case (IMS34_C50) {

        // MTSI and Message Waiting Indication

        v_ApplCond := pc_MultimediaTelephonyService and pc_MTSI_MessageWait;

      }

      case (IMS34_C55) {

        //  SM-over-IP sender

        v_ApplCond := pc_SMS_MO;

      }

      case (IMS34_C56) {

        // SM-over-IP receiver

        v_ApplCond := pc_SMS_MT;

      }

      case (IMS34_C57) {

        //MTSI AND communication waiting

        v_ApplCond := pc_MultimediaTelephonyService and pc_MTSI_CommWaiting;

      }

      case (IMS34_C58) {

        // IMS security AND Indicate Sigcomp

        v_ApplCond := pc_IMS_Sec and pc_IndicateSigcomp;

      }

      case (IMS34_C59) {

        // IMS emergency services

        v_ApplCond := pc_IMS_EmergencyCall;

      }

      case (IMS34_C62) {

        //IMS emergency services and emergency speech call

        v_ApplCond := pc_IMS_EmergencyCall and pc_EmergSpeech ;

      }

      case (IMS34_C63) {

        //IMS emergency services and speech and Communication Hold

        v_ApplCond := pc_IMS_EmergencyCall and pc_MTSI_Speech and pc_MTSI_CommHold;

      }

      case (IMS34_C64) {

        //IMS emergency services and EUTRA FDD or EUTRA TDD

        v_ApplCond := pc_IMS_EmergencyCall and (pc_eFDD or pc_eTDD);

      }

      case (IMS34_C65) {

        // IMS emergency services and UTRA and PS domain

        v_ApplCond :=pc_IMS_EmergencyCall and pc_GERAN and pc_PS;

      }

      case (IMS34_C66) {

        //Speech and IMS emergency call and emergency call over GERAN or UTRAN or 1XRTT

        v_ApplCond := pc_BidirecVoiceOverIMS and pc_MTSI_Speech and (pc_EmergSpeech or pc_1xRTT) ;

      }

      case (IMS34_C68) {

        //MTSI AND Initiating session AND MTSI speech AND E-UTRA

        v_ApplCond := pc_MultimediaTelephonyService and pc_InitiateSession and  pc_MTSI_Speech and (pc_eFDD or pc_eTDD);

      }

      case (IMS34_C69) {

        //MTSI AND Initiating session AND MTSI speech AND MTSI video AND E-UTRA

        v_ApplCond := pc_MultimediaTelephonyService and pc_InitiateSession and  pc_MTSI_Speech and pc_MTSI_Video and (pc_eFDD or pc_eTDD) ;

      }

      case (IMS34_C70) {

        //MTSI AND MTSI speech AND MTSI video AND MTSI Video H.264 CBP Level 1.2

        v_ApplCond := pc_MultimediaTelephonyService and pc_InitiateSession and pc_MTSI_Video and pc_VideoCodecH264CBP;

      }

      case (IMS34_C71) {

        //MTSI AND Initiating session AND MTSI speech AND MTSI video AND MTSI Video H.264 CBP Level 1.2

        v_ApplCond := pc_MultimediaTelephonyService and pc_InitiateSession and pc_MTSI_Speech and pc_MTSI_Video and pc_VideoCodecH264CBP;

      }

      case (IMS34_C72) {

        //IMS emergency services and MTSI and speech and Communication Hold during emergency call.

        v_ApplCond := pc_IMS_EmergencyCall and pc_MultimediaTelephonyService and  pc_MTSI_Speech and pc_CommunicationHold_DuringEmergencyCall;

      }

      case (IMS34_C73) {

        //IMS emergency services and MTSI and speech

        v_ApplCond := pc_IMS_EmergencyCall and pc_MultimediaTelephonyService and  pc_MTSI_Speech;

      }

      case else {

        v_ApplCond := false;

      }

    }

    return v_ApplCond;

  }

}

Change 2 – Renaming Condition Types in control part
	Function name
	control

	Reason for change
	See change 1.

	Summary of change
	Referenced the extended Condition Type names with the appropriate ATS-prefix in the control part of Testcase modules “IMS_IRAT” and “IMS_CC” (… and maybe also “LTE_EPS”).

	TTCN module
	IMS_IRAT_Testcases.ttcn         (… example shown below …)

IMS_CC_Testcases.ttcn            (… shall be done similarily …)

LTE_EPS_Testcases.ttcn          (… may be done similarily …)

	MCC160 Comment
	Not needed, see change 1


Before change

…
  //----------------------------------------------------------------------------
  control {
    if (f_SelectionExpr(C62))                                                      { execute(TC_19_1_3()); }
    if (f_SelectionExpr(C66))                                                      { execute(TC_19_3_2()); }
    if (f_SelectionExpr(C59))                                                      { execute(TC_19_4_1()); }
    if (f_SelectionExpr(C64))                                                      { execute(TC_19_4_2()); }
    if (f_SelectionExpr(C59))                                                      { execute(TC_19_4_5()); }
    if (f_SelectionExpr(C59))                                                      { execute(TC_19_5_1()); }
  }
}
After change

…
  //----------------------------------------------------------------------------
  control {
    if (f_SelectionExpr(IMS_IRAT_C62))                                                      { execute(TC_19_1_3()); }
    if (f_SelectionExpr(IMS_IRAT_C66))                                                      { execute(TC_19_3_2()); }
    if (f_SelectionExpr(IMS_IRAT_C59))                                                      { execute(TC_19_4_1()); }
    if (f_SelectionExpr(IMS_IRAT_C64))                                                      { execute(TC_19_4_2()); }
    if (f_SelectionExpr(IMS_IRAT_C59))                                                      { execute(TC_19_4_5()); }
    if (f_SelectionExpr(IMS_IRAT_C59))                                                      { execute(TC_19_5_1()); }
  }
}

